PHOTOMETRIC CALIBRATION OF THE IUE

X. Quantification of the High Dispersion Order Overlap Problem for SWP

L. Bianchi and R. Bohlin

I. INTRODUCTION

A long standing problem with IUE data is the determination of the true background level in the region of the spectral format where the high dispersion orders are closely spaced. Bianchi (1980) has outlined a technique for determining the net spectrum directly from the extracted gross data without reference to the extracted interorder background. The problem has been reduced, but not eliminated, by the introduction of the new software and automatic registration techniques (Bohlin and Turnrose 1982 and Thompson and Bohlin 1982.) These two production improvements have made high dispersion extractions consistent and stable, which is essential to the correction technique outlined here.

In order to evaluate any correction technique and to quantify the errors in IUE line profiles that are caused by the order overlap, we have compared line depths in IUE spectra to line depths observed by the *Copernicus* satellite. The excess line depth (i.e. order overlap) in IUE spectra can be expressed as a percent of the local net continuum level. The interpretation of these excesses suggest that the amount of order overlap for point sources is about 32% at 1150Å and decreases to zero at about 1400 Å. The transfer of a spectral feature from one order to the next is below the 5% level. The net is the most appropriate quantity to scale the order overlap, because the background and gross are affected by the radiation level and the variability of the camera null level.

II. COPERNICUS DATA

The resolution of *Copernicus* is 0.05Å for U1 spectra and 0.2Å for U2. Since the IUE resolution is 0.1Å, the absorption lines chosen for study are those that are broad enough so that the U1 and U2 line depths agree to within 5%. The choice of lines was further restricted to strong lines with central depths between 0 and 35% of the continuum. The spectra studied are those with complete U1 scans: zeta Oph(Morton 1975), zeta Pup(Morton and Underhill 1977), tau Sco(Rogerson and Upson 1977), and iota Her (Upson and Rogerson 1980). Except for iota Her (see section IV), all *Copernicus* U1 and U2 spectra were corrected for instrumental scattered light from the grating and for the U2 stray light by the method of Bohlin (1975). The *Copernicus* line depths should be accurate to about 3% of the continuum, although about half of the lines considered go to zero in both U1 and U2 and, therefore, have zero error.

III. IUE DATA

The four IUE spectra are listed in Table 1.

TABLE 1

HD	NAME	SWP NO.	APER.	EXP(s)
66811	zeta Pup	13726	L	4
149438	tau Sco	16222	L	6
149757	zeta Oph	14428	L	24
160762	iota Her	5720	S	80

All four spectra were reprocessed for this study with the production software in effect at GSFC in June 1982. The continua are drawn in the same way as for the Copernicus spectra, so that the main source of error is in the choice of the bottom of the line. Again, this error is estimated at 3%.

IV. RESULTS

The amount of order overlap in IUE spectra can be measured by the difference between the line depth in Copernicus and the line depth in IUE. The measured order overlap is shown in Fig. 1, as measured in % of the local net continuum. The first letter of the star's constellation is centered at each measured value. The vertical bars with the letter at the top of the bar represent the range of order overlap as determined by U1 and U2, independently. The straightline fit through the data of Fig. 1 is drawn without regard for the iota Her points which lie systematically low. Evidently, either the Copernicus scattered light corection is too large or this small aperture IUE spectrum has less order overlap. The former possibility is relevant, since Upson and Rogerson (1980) found that the normal scattered light correction is not applicable to the Copernicus U1 data for iota Her. Darius (1980) argued against any lower order overlap in the small aperture, but his large error bars permit the small ~5% difference found here. One error bar of ± 3% is illustrated, and all data (except iota Her) agree with the fit within their expected uncertainty.

Since the measured lines are all broad and deep, the contribution of the order itself to the excess background is generally small. Therefore, the actual order overlap appropriate to regions without strong lines in adjacent orders is significantly more than the amount shown in Fig. 1.(See Appendix 1.)



P - ζ Pup
O - ζ Oph
S - γ Sco
H - ι Her
Measured Order Overlap (% of NET Continuum)
C
λ (Å)

Fig. 1

V. ORDER OVERLAP IN THE OLD SOFTWARE

Since the background was systematically higher with the old high dispersion software in use before Nov. 10, 1981 at GSFC and before March 11, 1982 at Vilspa, the order overlap was worse. The amount of added order overlap in old software is the ratio of NEW/OLD net spectra as shown in Fig. 4 of Bohlin and Turnrose (1982). The maximum increase is ~10% at the shortest wavelengths and drops to zero near 1400Å, giving a maximum expected order overlap of ~42% for old SWP spectra.

However, an astronomer should regard the 42% derived above as a lower limit to the actual uncertainty in old reductions, because the extracted background was not accurately registered in a routine way before Nov. 24, 1981 at GSFC and March 11, 1982 at Vilspa (Thompson and Bohlin 1982). The random photometric errors before these dates can be as large as 35% (Bohlin and Coulter 1982). For problems that require accurate backgrounds, an astronomer shold have the spectra reprocessed with the modern system. The background extraction should continue to improve with the planned implementation of techniques to remove the residual geometric distortion in the orders.

VI. CORRECTION TECHNIQUES

If reprocessing of old reductions is impractical, a correction technique is recommended such as that of Bianchi (1980), which uses only the extracted gross spectrum. This technique assumes a Gaussian plus a Lorentz function for the order profile and takes into account the correction for the neighboring two orders. The accuracy of the result depends on actual shape of the long range wings of the PSF.

For modern reductions with proper spectral registration, the amount of order overlap measured in Fig. 1 can be used to make a simple, but sufficiently accurate, correction as described in Appendix 1.

As an example of this correction technique, Fig. 2 shows corrected orders (solid line) 114 to 111 in the spectrum of zeta Pup. The dashed line is the amount of the order overlap correction that has been added to the standard extracted net to get the solid line displayed. Both the Lα line and the shifted absorption component of the NV P-Cygni line go to zero in the *Copernicus* spectra. The Fortran subroutine used to compute the order overlap correction is in Appendix 2.


M=114
FN
1.00 E4
1205
1210
1215
1220
M=113
FN
1.00 E4
ORDER OVERLAP CORRECTION
CORRECTED NET
1215
1220
1225
1230
M=112
FN
1.00 E4
1225
1230
1235
1240
M=111
FN
1.00 E4
1235
1240
1245
1250
λ(Å)

Fig. 2

APPENDIX 1

The order overlap correction derived here deals with the order m itself and the neighboring two orders m-1 and m+1. The wings of all other orders should contribute nearly equal amounts to extracted gross and background spectra. The derivation here is for point sources in focus. Extended sources and out of focus high dispersion *IUE* spectra cannot be reduced properly with standard procedures at the short wavelengths where the orders are not well separated.

The corrected net N_o can be expressed in terms of the net n_o on the new software tapes as

$$N_o = n_o + \Delta B_o + \frac{\Delta B_- + \Delta B_+}{2} - \Delta N_- - \Delta N_+ . \qquad (1)$$

The subscripts -, o, and + refer to the orders m-1, m, and m+1, respectively throughout this discussion. The corrections ΔB are normalized to the extraction slit height and are due to the fact that the extracted background is too high which makes the net n_o too low. The corrections ΔN are the excess contributions to the gross from the wings of the adjacent orders. A correction ΔN_o is *not* needed for the few percent of the order m that lies outside the extraction slit, because a photometrically stable signal can be defined for any length slit as long as the extraction is precisely registered.

For the case of a deep line as measured by the correction C in Fig. 1, $\Delta B_o = 0$. On the average, the neighboring orders have approximately equal net continua n. With these assumptions and some knowledge of the order profile shape, a solution can be obtained. The precise PSF for *IUE* is not known, however, Bianchi (1980) has shown that the core of the profile is Gaussian with a longer range component in the wings. These wings produce the elevated background in the short wavelength orders and probably drop off as r^{-2}, where r is the distance from the peak of the order. De Boer, Preussner, and Grewing (1982) find *IUE* high dispersion profiles are purely Gaussian but suggest that their results are consistent with Bianchi, presumably because the faint Lorentz wings are difficult to detect.

Thus, if b is the background contribution due to one order, then this order contributes as an increase of b/4 to the neighboring net and as b/9 to the background on the other side of order m (See Fig. 3). In summary:

$$\Delta B_o = 0$$
$$\Delta B_- = \Delta B_+ = b + b/9$$
$$\Delta N_- = \Delta N_+ = b/4$$
$$n_- = n_+ = n$$



m-1
m
m+1
FN Perpendicular to the Orders
b
b/4
b/9
Locally Uniform Background
0
Distance Perpendicular to the Orders

Fig.3

Eq. 1 becomes:

$$N_o - n_o = b + b/9 - b/4 - b/4 = \frac{11b}{18} \qquad (2)$$

Since this difference $N_o - n_o$ is just what has been measured in Fig. 1, where C if the fractional correction in terms of n, Eq. 2 becomes:

$$b = \frac{18Cn}{11} \qquad (3)$$

This suggests for arbitrary continua in the three orders that the contribution to the background from any order is:

$$b_- = \frac{18C}{11} n_-$$

$$b_o = \frac{18Cn_o}{11}$$

$$b_+ = \frac{18Cn_+}{11}$$

For the case of non-zero signal in the order m, the contribution

$$\Delta B_o = b_o$$

and the general solution to Eq. 1 becomes

$$N_o = n_o + \frac{18C}{11}[n_o] + \frac{10C}{11}[n_-] + \frac{10C}{11}[n_+] - \frac{9C}{22}[n_-] - \frac{9C}{22}[n_+]$$

$$N_o = n_o + 1.636C\,[n_o] + 0.5C\,[n_-] + 0.5C\,[n_+] \qquad (4)$$

where the [ ] indicates the appropriately smoothed net spectrum. The appropriate smoothing is 31 points done twice, just the same as the background smoothing for the new software, since the correction is essentially for errors in the smooth background that is used to compute the net on the tape. The FORTRAN program to implement Eq. 4 appears in Appendix 2.

In the case where the 3 continua are all equal, Eq. 4 becomes:

$$\frac{N_o - n_o}{[n_o]} = 2.636C \quad , \qquad (5)$$

which is used to estimate the maximum order overlap of 32% when C = 0.12 at 1150Å.

Order overlap is primarily an artificially raised background caused by the overlapping wings of the order profile in the higher echelle orders. A secondary effect of order overlap is the transfer of a spectral feature from one order to a neighboring order, since the wings of the point spread function (PSF) are not zero at the location of a close order. The depth of these "ghost lines" is the contribution ΔN_+ from a neighboring order. The smoothing of these ΔN terms was assumed to be the large interval of 31 points in deriving Eq. 4. The following arguments will set a low limit on the importance of ghost lines and therefore justify the large smoothing of these excess contributions to the extracted net: Suppose that a zero depth line in the continuum of m + 1 with $n_+ = n_o$ is present as a narrow ghost line in order m. The depth of the dip is

$$\frac{\Delta N_+}{n_o} = \frac{b}{4n_o} = 0.41C$$

This is a maximum of 5% at 1150Å and drops to the even more insignificant upper limit of 3% longward of 1260Å. The lack of any visible ghost lines from the strong emission lines in WAVECAL spectra is consistent with these limits.

APPENDIX 2

```
      SUBROUTINE OVRLAP(NUMORD,MPTS,WLM,EPS,FNET,OOCOR)
C+
C     CORRECT THE SWP (ICAM=3) IUE HI-DISP SPECTRA FOR ORDER OVERLAP.
C
CU    USER INPUT IS THE NUMBER OF ORDERS (NUMORD), THE NUMBER OF POINTS
CU    IN EACH ORDER (MPTS), THE WAVELENGTHS OF EACH POINT IN THE ORDER
CU    (WLM), AND THE EPSILON VALUES (EPS).
CU    THE PROGRAM COMPUTES THE CORRECTION (OOCOR) AND ADDS
CU    IT TO FNET.  FNET IS FLAGGED AS A BAD POINT (-1.E-20) IF THE
CU    POINT IS A RESEAU, SATURATED, OR A PING. THE CORRECTED ORDERS
CU    ARE 2 THROUGH NUMORD-1.
C
CS    SUBROUTINE AV IS NEEDED TO SMOOTH WITH A BOX FILTER (31 POINTS
CS    WIDE) AND ARRAY LENGTH NPTS.
C
CH    AUTHOR: RALPH BOHLIN
CH            GSFC-CODE 681 AND ST SCI HOMEWOOD CAMPUS
CH    WRITTEN: 3 JULY 1982
C-
      COMMON/NPTS/NPTS
      COMMON/VICHDR/IL,IS,ICAM
      DIMENSION MPTS(NUMORD),WLM(1022,NUMORD),EPS(1022,NUMORD),
     C        FNET(1022,NUMORD),OOCOR(1022,NUMORD),AVG(1022,3),
     C        AVGT(1022),CORR(3)
      IF(ICAM.NE.3)RETURN
C
C     INITIALIZE THE 1ST 2 SMOOTHED ORDERS & CORRECTION FACTORS.
          DO 100 I=1,2
          NPTS=MPTS(I)
          CALL AV(FNET(1,I),AVGT,31)
          CALL AV(AVGT,AVG(1,I),31)
          WLCOR=WLM(NPTS/2,I)
100       CORR(I)=(1400.-WLCOR)*0.000533/2.
C
C     MAIN LOOP WHERE VARIABLE ENDING WITH: LO-PREVIOUS ORDER
C                                          AT-ORDER BEING CORRECTED
C                                          HI-NEXT ORDER
          LSTCOR=NUMORD-1
          DO 300 I=2,LSTCOR
          ILO=MOD(I-2,3)+1
          IAT=MOD(I-1,3)+1
          IHI=MOD(I  ,3)+1
          NPTS=MPTS(I+1)
          CALL AV(FNET(1,I+1),AVGT,31)
          CALL AV(AVGT,AVG(1,IHI),31)
          WLCOR=WLM(NPTS/2,I+1)
          CORR(IHI)=(1400.-WLCOR)*0.000533/2.
          MIDLO=MPTS(I-1)/2
          MIDAT=MPTS(I  )/2
          MIDHI=MPTS(I+1)/2
              NPTS=MPTS(I)
              DO 200 J=1,NPTS
              NPTLO=J-MIDAT+MIDLO
              NPTHI=J-MIDAT+MIDHI
              IF(NPTLO.LT.1)NPTLO=1
              IF(NPTHI.LT.1)NPTHI=1
              IF(NPTLO.GT.MPTS(I-1))NPTLO=MPTS(I-1)
              IF(NPTHI.GT.MPTS(I+1))NPTHI=MPTS(I+1)
COMPUTE CORRECTION DUE TO NEIGHBORING ORDERS
              OOCOR(J,I)=AVG(NPTLO,ILO)*CORR(ILO)
     C                   +AVG(NPTHI,IHI)*CORR(IHI)
CORRECT FOR NEIGHBORING ORDERS. COMPUTE CORR. FOR ORDER ITSELF & ADD IT
           OOCOR(J,I)=OOCOR(J,I)+(AVG(J,IAT)+OOCOR(J,I))*3.273*CORR(IAT)
              FNET(J,I)=FNET(J,I)+OOCOR(J,I)
              IF(EPS(J,I).LE.-220.)FNET(J,I)=-1.E-20
200           CONTINUE
300       CONTINUE
      RETURN
      END
```

REFERENCES

Bianchi, L. 1980, in IUE Data Reduction, ed. W. W. Weiss et al., Vienna: Austrian Solar and Space Agency, p. 161.

Bohlin, R. C. 1975, Ap.J., 200, 402.

Bohlin, R. C. and Coulter, B. T. 1982, "Photometric Calibration of the IUE IX. Photometric Stability in High Dispersion," NASA IUE Newsletter No. 19, p. 48.

Bohlin, R. C., and Turnrose, B. E. 1982, "IUE data Reduction XXV. Implementation of Basic Improvements to Extraction of High Dispersion Spectra," NASA IUE Newsletter No. 18,.p. 29.

Darius, J. 1980, "Spectral Resolution and Interorder Background," ESA IUE Newsletter No. 5, p. 28.

de Boer, K. S., Preussner, P.-R., and Grewing, M. 1982, Astron. and Astrophys., in press.

Morton, D. C. 1975, Ap.J., 197, 85.

Morton, D. C., and Underhill, A. B. 1977, Ap.J. Suppl., 33, 83.

Rogerson, J. B., and Upson, W. L. 1977, Ap.J. Suppl., 35, 37.

Thompson, R. W., and Bohlin, R. C. 1982, "IUE Data Reduction XXVI. Automatic Registration of the Extraction Slit with the Spectral Format," NASA IUE Newsletter No. 18, p.45.

Upson, W. L., and Rogerson, J. G. 1980, Ap. J. Suppl., 42, 175.